



# *Polyaxial Stress Analysis of Underground Openings using FLAC<sup>®</sup>*

*D. Scussel\**  
*S. Chandra\*\**

*\*School of Civil Engineering, Surveying and Construction  
University of Kwazulu Natal, Durban, South Africa  
Email: scussel@ukzn.ac.za*

*\*\*Department of Civil Engineering  
Indian Institute of Technology, Kanpur 208 016, India*

## **ABSTRACT**

The traditional design methodologies for tunnel and underground excavations are divided in to three categories: Empirical approaches, Analytical approaches, and Observational approaches, whereas in the last years the Numerical approach has strongly become popular both for the intrinsic simplicity of the software packages and their ability to manage problems unsolvable with the classic methods.

In this paper, the underground openings have been analysed using constitutive models other than the Mohr Coulomb's theory. FLAC is used for the analysis and the software has been implemented to include the polyaxial strength criterion. The details of the modifications made in the software are presented and the results are compared with the Singh's elasto-plastic stress distribution in squeezing grounds. This study will develop better comprehension of the behaviour of the underground openings and also provide a useful tool to the designers in the planning stages.

**Keywords:** Tunnel excavation; Polyaxial strength criterion; FLAC, Strain softening; FDM; Finite difference analysis.

## **1. INTRODUCTION**

Many constitutive models have been developed to describe the behaviour of a rock mass after modification of its equilibrium. In severe conditions none of these, however, has demonstrated sufficient correlation to the effective measured reactions. In fact, several experiences of back analysis in tunnels excavation (Jethwa, 1981), when compared to the results of the more applied designing procedures, have shown a marked tendency to overestimate the squeezing of the rock masses.

To meet the needs of a more suitable theory for squeezing conditions, Wang and Kemeny (1995) performed several tests on anisotropic tuff to advance the hypothesis that the intermediate principal stress in an anisotropic rock mass under a polyaxial stress field could influences its behavior.




Fig. 1 - Stress distribution around and underground opening

Rock mass in the vicinity of an underground excavation is a clear example of a medium subjected to a polyaxial stress field:  $\sigma_3$  is very small or equal to zero,  $\sigma_2$  is close to the in situ vertical stress (for deep tunnels) and  $\sigma_1$  could be double the intermediate principal stress. Moving away in the radial direction, the difference between maximum and minimum stress is less appreciable.

The classical strength theory assumes that only minor and major principal stresses influence the stability of the rock surrounding the excavation. However, in practical situations, the consideration of intermediate principal stress results in enhancement of strength. This can be explained by the significant work done by the intermediate principal stress component along the tunnel direction that compresses wedges of rock, increasing their global resistance and preventing rock falls.

## 2. ELASTO-PLASTIC THEORY OF STRESS DISTRIBUTION IN BROKEN ZONE USING POLYAXIAL STRENGTH CRITERION

Singh et al. (1998) investigated the effects of the intermediate principal stress on the strength of anisotropic rock mass, and proposed to modify the Mohr-Coulomb's criterion by replacing  $\sigma_3$  with the average value of  $\sigma_2$  and  $\sigma_3$ . The polyaxial strength criterion based on semi-empirical approach has shown better correlation between analytical results and observations. The criterion suggested by them is given below:

$$\sigma_1 - \sigma_3 = \sigma_{ci} + \left( \frac{\sigma_2 + \sigma_3}{2} \right) \frac{2 \sin \phi}{1 - \sin \phi} \quad (1)$$

Starting from Equation (1), they formulated an elasto-plastic theory of stress distribution in broken zone in squeezing ground.

Initial hypothesis can be summarized as follows (Fig. 2):

- Rock mass is isotropic, homogeneous and dry;
- Rock mass follows the polyaxial strength criterion in the elastic zone, whereas the Mohr-Coulomb's theory inside the broken zone;
- Circular tunnel of radius  $r_1$  is uniformly supported, and circular broken zone is of radius  $r_p$ ; and
- There is no rock burst or brittle failure.




Fig. 2 - Schematic boundary conditions of the problem

Stress distribution within the broken zone is given as:

$$\sigma_r = \left( P_b + \frac{q_{cr}}{\alpha} + \frac{\gamma}{1-\alpha} \right) \left( \frac{r}{r_p} \right)^\alpha - \frac{q_{cr}}{\alpha} \quad (2)$$

$$\sigma_\theta = q_{cr} + (1+\alpha)\sigma_r \quad (3)$$

Squeezing Pressure at the lining in the vertical direction ( $\theta = 90^\circ$ ):

$$P_v = \left( P_b + \frac{q_{cr}}{\alpha} + \frac{\gamma}{1-\alpha} r_p \right) \left( \frac{r_i}{r_p} \right)^\alpha - \frac{q_{cr}}{\alpha} - \frac{\gamma}{1-\alpha} r_i \quad (4)$$

and in the horizontal direction ( $\theta = 0^\circ$ ):

$$P_h = \left( P_b + \frac{q_{cr}}{\alpha} \right) \left( \frac{r_i}{r_p} \right)^\alpha - \frac{q_{cr}}{\alpha} \quad (5)$$

Squeezing pressure at the lining for hydrostatical initial stress and negligible effect of rock mass weight is:

$$P_i = \left( \frac{2P_v - q_{cmass} - \sigma_z A/2}{2 + A/2} + \frac{q_{cr}}{\alpha} \right) \left( \frac{r_i}{r_p} \right)^\alpha - \frac{q_{cr}}{\alpha} \quad (6)$$

Where

$$A = \frac{2 \sin \phi_p}{1 - \sin \phi_p}, \quad \alpha = \frac{2 \sin \phi_r}{1 - \sin \phi_r}, \quad q_{cmass} = 7\gamma Q^{1/3}, \quad q_{cr} = \frac{2c_r \cos \phi_r}{1 - \sin \phi_r} \quad \text{and} \quad P_b = P_v \frac{((1+\lambda) + 2(1-\lambda)\cos 2\theta) - q_{cmass} - \sigma_z A/2}{2 + A/2}$$

### 3. NUMERICAL ANALYSIS

#### 3.1 Statement of the Problem

The introduction of several engineering numerical analysis suites (FEM, FDM, BEM, DEM) has changed the approach to excavation problems. Now, it is possible to carry out a more detailed analysis considering complexities such as the influence of new parameters, particular geometries and boundaries shapes, introducing new excavation technique or considering the complex rock mass-liner interaction. Many very powerful codes have been developed for different constitutive models and are available for the analysis of geo-mechanical problems, but none of them consider the effect of the intermediate principal stress in the evaluation of plasticity.

The scope of this work is to include the polyaxial criterion among the more common constitutive model codes for FLAC and make it available for practical tunnel design. FLAC (Itasca) is a two-dimensional explicit finite difference program for solving many computational problems of geotechnical engineering and rock mechanics.

In this study FLAC has been chosen for implementing a user-defined constitutive model, which is not present in this software and also in any other standard software. The model has been compiled in FISH (Appendix B), the built in program language. The inclusion of this feature in this software makes it ideal software for many practical studies of underground openings.

#### 3.2 Modifications to Implement the Polyaxial Constitutive Model

To develop a FISH code to incorporate the polyaxial strength criterion, it is important to redefine the constitutive model's formulation consistent with the sign convention in FLAC. Starting with this, compression is taken as negative and the ordering of the principal stresses is  $\sigma_1 < \sigma_2 < \sigma_3$  as in structural engineering.

To avoid misunderstanding when the requested data for the execution of the model are inserted by not an expert user, the variable needed are always positive in sign. This obviously affects the formulation of the constitutive model because, in such a reference system, Cohesion and Uniaxial Compressive Strength would be negatives.

$$\sigma_1 - \sigma_3 - \sigma_{ci} - \left( \frac{\sigma_2 + \sigma_3}{2} \right) A = 0 \quad \Rightarrow \quad \sigma_1 - \sigma_3 + \sigma_{ci} - \left( \frac{\sigma_2 + \sigma_3}{2} \right) A = 0$$

Incorporating these, Mohr's Theory and polyaxial criterion will take the form given as:

$$\text{Mohr's:} \quad \sigma_1 - N_\varphi \sigma_3 + \sigma_{ci} = 0 \quad (7)$$

$$\text{Polyaxial:} \quad \sigma_1 - \sigma_3 + \sigma_{ci} - \left( \frac{\sigma_2 + \sigma_3}{2} \right) A = 0 \quad (8)$$

Other necessary step is to reformulate the polyaxial strength criterion to make it similar to the Mohr's formulation, as suggested below:

$$\sigma_1 - \sigma_3 N'_\phi + \sigma'_{ci} = 0 \quad (9)$$

Where  $\sigma_{ci}$  is the Compressive strength at the internal boundary of a tunnel

$$\sigma'_{ci} = \sigma_{ci} - \frac{\sigma_2}{2} A = 0$$

and, where 
$$N_\phi = \frac{1 + \sin \phi}{1 - \sin \phi} \quad N'_\phi = \frac{1}{1 - \sin \phi}$$

The polyaxial strength criterion as given by Equation 9 is shown in a graphical form in Figure 3. The similarity with the Mohr-Coulomb's criterion is quite evident. However, the polyaxial criterion incorporates the intermediate principal stress, which is not incorporated in Mohr-Coulomb's criterion. Therefore, the parameters  $N'_\phi$  and  $\sigma'_{ci}$  are different from the corresponding  $N_\phi$  and  $\sigma_{ci}$ . It is appropriate to highlight that the intermediate principal stress required to be evaluated only in elastic condition and is automatically computed by FLAC.




Fig. 2 - Graphical Polyaxial Constitutive Model

In order to have a three dimensional constitutive model in elastic zone and a bi-dimensional model after it fails, in the analysis, a new approach is suggested in this paper. A new relationship based on the similarity between Equations 7 and 9, is proposed below:

$$\sigma_1 - \sigma_3 \left( \frac{1 + \sin \alpha}{1 - \sin \phi} \right) + \sigma_{ci} - \sigma_2 \frac{\sin \beta}{1 - \sin \phi} = 0 \quad (10)$$

The advantage of the above equation is that it can handle both elastic zone and plastic zone by appropriately choosing the parameters. The values of the parameters  $\alpha$ ,  $\beta$ ,  $\phi$  and  $\sigma_{ci}$  to be used are given in Table 1.

Substituting in Equation 10 the peak and residuals values results in:

$$\sigma_1 - \sigma_3 \left( \frac{1}{1 - \sin \phi_p} \right) + 7\gamma Q^{1/3} - \sigma_2 \frac{\sin \phi_p}{1 - \sin \phi_p} = 0 \quad (11)$$

and

$$\sigma_1 - \sigma_3 \left( \frac{1 + \sin \phi_r}{1 - \sin \phi_r} \right) + \frac{2 \cos \phi_r}{1 - \sin \phi_r} = 0 \quad (12)$$

Table 1 - Values proposed for Peak and Residual State

	Peak	Residual
$\phi$	$\phi_p$	$\phi_r$
$\alpha$	0	$\phi_r$
$\beta$	$\phi_p$	0
$\sigma_{ci}$	$7\gamma Q^{1/3}$ or $\frac{2c_p \cos \phi_p}{1 - \sin \phi_p}$	$\frac{2c_r \cos \phi_r}{1 - \sin \phi_r}$

In this formulation the value of UCS is directly put as a parameter. The advantage of inputting the value directly is that it can be modified in the formulation without changing the FISH code every time. This way, the problem can be solved using the peak and residual parameters that could describe the behaviour of an elastic rock mass and transform the formulation of the failure criteria at plasticity.

### 3.3 Solution Scheme

In each step to compute the stresses, these are evaluated, transformed in principal stresses and ordered.

FLAC chooses a guess elastic strain increment and calculates the corresponding stress increments applying the Hooke's Law. The incremental stresses are given below:

$$\begin{cases} \Delta \sigma_1 = \alpha_1 \Delta e_1^e + \alpha_2 (\Delta e_2^e + \Delta e_3^e) \\ \Delta \sigma_2 = \alpha_1 \Delta e_2^e + \alpha_2 (\Delta e_1^e + \Delta e_3^e) \\ \Delta \sigma_3 = \alpha_1 \Delta e_3^e + \alpha_2 (\Delta e_1^e + \Delta e_2^e) \end{cases} \quad (13)$$

where  $\alpha_1 = K + 4G/3$  and  $\alpha_2 = K - 2G/3$


After this step, it is evaluated whether the new three components violate the yield criterion given by Equations 14 and 15 for shear or tension.

$$\text{Shear Yield function } f_s = \sigma_1 - N'_\phi \sigma_3 + \sigma'_{ci} \quad (14)$$

$$\text{Tension Yield function } f_t = \sigma^t - \sigma_3 \quad (15)$$

The equation for the bisector of the angle originated by the tension and the shear yield function is given as:

$$h(\sigma_1, \sigma_2, \sigma_3) = \sigma_3 - \sigma^t - \alpha^p (\sigma_1 - \sigma^p) \quad (16)$$

Fig. 3 - Domains for a specific  $\sigma_2$  value

where

$$\sigma^p = N'_\phi + \sqrt{1 + (N'_\phi)^2} \quad \text{and} \quad \alpha^p = N'_\phi \sigma' - \sigma'_{ci}$$

Equations 14, 15 and 16 are shown in a graphical form in Figure 4 in a two dimensional plot between  $\sigma_1$  and  $\sigma_3$  for a particular value of  $\sigma_2$ . A three dimensional plot is actually needed to represent the zones of failure or no failure but for the sake of simplicity a two dimensional plot is shown. By locating a stress point on this figure one can make out in which zone it lies. If the sign of  $f_s$  and  $h$  are negative, shear failure takes place, when  $h$  is positive and  $f_t$  negative, tensile failure takes place and no failure when both  $h$  and  $f_s$  are positive.

The violation of the yield criterion means that FLAC calculated a point beyond the yield function and plastic deformation takes place ( $e^p > 0$ ). A correction is needed to move it back to the yield boundary (the guess elastic strain increment was not elastic).

The treatment of the tensile failure is the same as in Mohr-Coulomb constitutive model in FLAC. Therefore only shear failure correction is applied in the present study.

Starting from the flow rule's formulation given as:

$$\Delta e_i^p = \lambda^s \frac{\partial g_s}{\partial \sigma_i} \quad (17)$$

$$g_s = \sigma_1 - N_\psi \sigma_3 \quad (18)$$

Where  $g_s$  is the Shear potential function and  $\lambda^s$  as unknown. In this function  $\sigma_2$  is absent because, although not constant in the entire domain, it is a constant value for a particular zone. The elastic guess increments in the three directions are given as:

$$\begin{cases} \Delta e_1^p = \lambda^s \\ \Delta e_2^p = 0 \\ \Delta e_3^p = -N_\psi \lambda^s \end{cases} \quad (19)$$

$$\text{where } N_{\psi} = \frac{1 + \sin \psi}{1 - \sin \psi}$$

The total increment applied at the beginning must be separated from its plastic part, which is calculated with the flow rule (Equation 17) and then substituted in Equation 21 that gives the value of elastic strain in the incremental expression of the Hooke's law to be used for computing the principal stresses (Equation 22):

$$\Delta e_i = \Delta e_i^e + \Delta e_i^p \quad i=1,2,3 \quad (20)$$

$$\Delta e_i^e = \Delta e_i - \Delta e_i^p \quad (21)$$

$$\begin{cases} \Delta \sigma_1 = \alpha_1 \Delta e_1 + \alpha_2 (\Delta e_2 + \Delta e_3) - \lambda^s (\alpha_1 - \alpha_2 N_{\psi}) \\ \Delta \sigma_2 = \alpha_1 \Delta e_2 + \alpha_2 (\Delta e_1 + \Delta e_3) - \lambda^s \alpha_2 (1 - N_{\psi}) \\ \Delta \sigma_3 = \alpha_1 \Delta e_3 + \alpha_2 (\Delta e_1 + \Delta e_2) - \lambda^s (-\alpha_1 N_{\psi} + \alpha_2) \end{cases} \quad (22)$$

The stress increment values are used to compute the new values of stresses as given by the equation given below:

$$\begin{cases} \sigma_1^{New} = \sigma_1^I - \lambda^s (\alpha_1 - \alpha_2 N_{\psi}) \\ \sigma_2^{New} = \sigma_2^I - \lambda^s \alpha_2 (1 - N_{\psi}) \\ \sigma_3^{New} = \sigma_3^I - \lambda^s (-\alpha_1 N_{\psi} + \alpha_2) \end{cases} \quad (23)$$

The superscript *New* means new values and I is used to represent the principal stresses obtained by adding the guess elastic strain to the initial principal stresses field. The second part of the expression is the stress component due to the plastic strain correction.

The value of  $\lambda^s$  now can be computed by using the new values of stresses using Equation 23 and substituted in shear yield function given by Equation 14. The right hand side of the equation is equated to zero to ensure that the point lies on the shear failure yield line, since the point cannot lie above that. The value of  $\lambda^s$  is obtained as:

$$\lambda^s = \frac{f_s(\sigma_i^I)}{\alpha_1 - \alpha_2 N_{\psi} - N_{\phi} (-\alpha_1 N_{\psi} + \alpha_2)} \quad (24)$$

Now FLAC can compute the new stress field and repeat again until the value of the maximum unbalanced force of the system reduces to a negligible value and thus a static solution is obtained.

### 3.4 Comparison of Numerical and Analytical Results

In the present study an example is considered for which the geometrical configuration is shown in Fig. 5. The numerical values of various parameters used for this problem are presented in Table 2. This example has been used to validate the implementation of the polyaxial constitutive model in the finite difference code in FLAC. The results of



this study for this particular example are compared with the results obtained by Elasto-Plastic theory of stress distribution in broken zone zone in squeezing ground conditions as suggested by Singh et al. (2006).




Fig. 4 - Geometrical configuration of the model

Table 2 - Data of the numerical problem

$r_i$	3.2 [m]	Q	0.001
distance of boundaries	$32 * r_i$ [m]	$\gamma$	27 [kN/m <sup>3</sup> ]
$\phi_p$	30 [°]	$P_v$	15 [MPa]
$\phi_r$	20 [°]	$P_i$	1.52 [MPa]
$c_p$	2 [MPa]	$q_{emass}$	1.89 [MPa]
$c_r$	0.1 [MPa]	$\lambda$	1
E	5 [GPa]		
$\nu$	0.25		

The analytical solution of the above problem has been obtained with the spreadsheet presented in Appendix A.

Results of analytical and numerical stress distribution computation are shown in Fig. 6.




Fig. 5 - Comparison of analytical and numerical FLAC solution for the tunnel problem in Fig. 3

The figure shows generally good agreement between the results obtained by two methods. It can be observed that the predictions are very good up to a certain distance and beyond a certain distance. There is some problem at the boundary of the plastic zone where the predictions cannot be made at all due to the inherent drawbacks of using finite difference method and FLAC. The mesh could not be refined further in this region due to the limitation of the software. The difference in the stresses computed by two different methods at various distances in percentages is reported in appendix in Table A-1.

#### 4. CONCLUSIONS

The elasto-plastic polyaxial model produced in this paper introduces an alternative to design tunnels in squeezing rock masses. A new relationship between the principal stresses and uniaxial compressive strength is suggested in this work, which can handle both the elastic and plastic zone according to Singh's Theory by appropriately choosing the parameters. The proposed relationship is used to bring out the effect of intermediate principal stress as the development of principal stresses in an underground opening. Through an example it is shown that the effect of the intermediate principal stress contributes to the enhancement of the peak characteristics of the underground excavation.

## Acknowledgments

Professor Mahendra Singh, Indian Institute of Technology Roorkee, reviewed the initial draft of the paper. The Authors are grateful for his suggestions.

## List of Symbols

$r, r_i, r_p$ :	Distance from the centre of the tunnel, internal and plastic radius;
$e^e, e^p, e$ :	Elastic and Plastic Strain
$K, G, \nu, E$ :	Bulk, Shear, Poisson's, Young's modulus;
$Q$ :	Barton's rock mass quality ;
$\gamma$ :	Rock mass unit weight (g/cc);
$\sigma_1, \sigma_2, \sigma_3$ :	Maximum, Intermediate, Minimum principal stress;
$\sigma^t$ :	Tension Cut;
$\sigma_r, \sigma_\theta, \tau_{\theta\rho}, Pz$ :	Stress distribution around a tunnel in Radial, Tangential directions and along the Tunnel direction;
$P_v, P_h, \lambda$ :	Overburden pressure, Horizontal pressure, Horizontal ratio;
$q_{cmass}, q_{cr}, \sigma_{ci}$ :	Peak, Residual, Uniaxial compressive strength;
$\phi_p, \phi_r, \phi$ :	Peak, Residual, internal friction angle;
$c_p, c_r, c$ :	Peak, Residual, Cohesion;
$\psi$ :	Dilation angle;
$\theta$ :	Angle between the horizontal axis of a tunnel and the line between its centre and point considered.

## References

- Barla, G. (1995). Squeezing rocks in tunnels, ISRM News Journal, 2(3&4), pp.44-49.
- Barla, G. (2004). Tunnelling Under Squeezing Rock Conditions, [www.polito.it/ricerca/rockmech/publcazioni/art-rivista](http://www.polito.it/ricerca/rockmech/publcazioni/art-rivista).
- Daemen, J.J.K. (1975). Tunnel Support Loading caused by Rock Failure. PhD thesis, University of Minnesota, Minneapolis, U.S.A.
- Itasca (2006), Reference Manual, FLAC 5.0, Itasca Consulting Group Inc, Minneapolis (2006).
- Jethwa, J. L. (1981). Evaluation of Rock Pressures in Tunnels through Squeezing Ground in Lower Himalayas, PhD thesis, Department of Civil Engineering, University of Roorkee, India, p.272.
- Ribacchi, R., (1986). Stato di sforzo e deformazione intorno ad una galleria, LXI, pp. 135-184.
- Singh, Bhawani, Goel, R. K., Mehrotra, V. K., Garg, S. K. and Allu, M. R. (1998). Effect of intermediate principal stress on strength of anisotropic rock mass, Tunnelling & Underground Space Technology, Pergamon, 13(1), pp.71-79.
- Singh, Bhawani, Jethwa, J. L., Dube, A. K. and Singh, B. (1992). Correlation between observed support pressure and rock mass quality. Tunnelling & Underground Space Technology, Pergamon, 7(1), pp.59-74.
- Singh, B., Goel, R.K. (2006). Tunnelling in Weak Rocks. Geo-Engineering, 5, Elsevier Science, p. 512.
- Singh, M., Singh, Bhawani, Choudhari, J. (2007). Critical strain and squeezing of rock mass in tunnels, Tunnelling and Underground Space Technology, 22, pp.343 - 350.
- Wang, R. and Kemeny, J. M. (1995). A new empirical failure criterion under polyaxial compressive stresses. Rock Mechanics: Proc. 35th U. S. Symposium, Reprinted from: Eds: J. J. K. Daemen and R. A. Schultz, Lake Tahoe, 4-7 June 1995, p.950.

## APPENDIX A: SPREADSHEET FOR THE IMPLEMENTATION OF THE TUNNEL SOLUTION

Geometrical Data and Strength Parameter				Determination of Pi				
$\gamma$	27.0	kN/m <sup>3</sup>	$r_i$	1.5	m	$P_b$	4.37	MPa
Q	0.001		$r_p$	8.0	m	$P_i$	1.52	MPa
$\phi_p$	30.0	°	$P_v$	15.0	MPa			
$\phi_r$	20.0	°	$P_z$	15.0	MPa			
$c_p$	2.0	MPa	$\lambda$	1.0				
$c_r$	0.1	MPa						

Extract from the spreadsheet for the implementation of the stress distribution around circular openings subjected to symmetrical loading:

Table A-1

r	Flac		Theoretical		Difference	
	$\sigma_r/P_v$	$\sigma_t/P_v$	$\sigma_r/P_v$	$\sigma_t/P_v$	Radial	Tangential
3.59	0.1186	0.2609	0.1161	0.2559	2.1162%	1.9434%
4.36	0.1497	0.3236	0.1465	0.3179	2.1755%	1.8026%
5.15	0.1807	0.3840	0.1774	0.3809	1.8515%	0.8141%
5.94	0.2125	0.4524	0.2088	0.4450	1.7605%	1.6720%
6.74	0.2453	0.5193	0.2407	0.5099	1.9201%	1.8371%
7.55	0.2813	0.5927	0.2730	0.5759	3.0231%	2.9087%
8.36	0.3586	1.6500	0.3512	1.6488	2.1048%	0.0733%
9.19	0.4706	1.5440	0.4624	1.5376	1.7746%	0.4159%
9.94	0.5497	1.4660	0.5406	1.4594	1.6849%	0.4517%
10.62	0.6071	1.4100	0.5979	1.4021	1.5448%	0.5609%
11.30	0.6546	1.3620	0.6448	1.3552	1.5188%	0.5022%
11.99	0.6946	1.3230	0.6845	1.3155	1.4738%	0.5710%
12.69	0.7284	1.2890	0.7184	1.2816	1.3980%	0.5740%
13.40	0.7572	1.2600	0.7474	1.2526	1.3096%	0.5918%
14.11	0.7821	1.2350	0.7722	1.2278	1.2830%	0.5858%
14.83	0.8036	1.2140	0.7938	1.2062	1.2377%	0.6446%
15.56	0.8224	1.1950	0.8127	1.1873	1.1970%	0.6462%
16.29	0.8389	1.1780	0.8291	1.1709	1.1838%	0.6051%
17.04	0.8535	1.1640	0.8438	1.1562	1.1496%	0.6746%
17.79	0.8664	1.1510	0.8567	1.1433	1.1332%	0.6728%
18.54	0.8778	1.1400	0.8681	1.1319	1.1230%	0.7113%
19.31	0.8882	1.1290	0.8784	1.1216	1.1197%	0.6567%
20.08	0.8973	1.1200	0.8875	1.1125	1.1025%	0.6755%
			<b>Average</b>		<b>1.573%</b>	<b>0.895%</b>

## APPENDIX B: FISH Code

```

* -----
* FISH version of Singh model with
* strain hardening/softening
* -----
set echo off
def b_singh
  constitutive_model
  f_prop m_g m_k m_fric m_dil m_ten
  f_prop m_beta m_delta m_q
  f_prop m_ftab m_ttab m_ind m_epdev m_epten
  f_prop m_btab m_dtab m_qtab
  f_prop m_e1 m_e2 m_x1 m_sh2
  f_prop m_npsi m_nphi m_csnp m_qdelta
;
float $sphi $spsi $s1i $s22i $s12i $s33i $sdif $s0 $rad $s1 $s2 $s3
float $sii $sdi $fs $salam $ft $salamt $cs2 $si2 $dc2 $dss
float $sdelta $sbeta
float $sapex $epsav $psav $de1ps $de3ps $depm $eps $ept $seps
float $bisc $pdiv $anphi $stco
int $scase $sm_err $iftab $ittab
int $ibtab $iqtab $idtab
;
Case_of mode
; -----
; Initialisation section
; -----
Case 1
; --- put initial table values in prop arrays ----
if m_epdev = 0.0 then
  if m_epten = 0.0 then
    $iftab = int(m_ftab)
    $idtab = int(m_dtab)
    $ittab = int(m_ttab)
    $ibtab = int(m_btab)
    $iqtab = int(m_qtab)

    if $iftab # 0 then
      m_fric = table($iftab, 0.0)
    end_if
    if $idtab # 0 then
      m_delta = table($idtab, 0.0)
    end_if
    if $ittab # 0 then
      m_ten = table($ittab, 0.0)
    end_if
    if $ibtab # 0 then
      m_beta = table($ibtab, 0.0)
    end_if
    if $iqtab # 0 then
      m_q = table($iqtab, 0.0)
    end_if
  end_if
end_if
; --- data check ---
$sm_err = 0
if m_fric > 89.0 then
  $sm_err = 1
end_if
if m_ten < 0.0 then
  $sm_err = 2
end_if
if m_beta < 0.0 then
  $sm_err = 3
end_if
if m_delta < 0.0 then
  $sm_err = 4
end_if
if $sm_err # 0 then
  nerr = 126
  error = 1
end_if
;
$sphi = sin(m_fric * degrad)
$sdelta = sin(m_delta * degrad)
$sbeta = sin(m_beta * degrad)
$spsi = sin(m_dil * degrad)
m_npsi = (1.0 + $spsi) / (1.0 - $spsi)
m_nphi = (1.0 + $sbeta) / (1.0 - $sbeta)
m_qdelta = $sdelta / (1.0 - $sphi)
m_csnp = m_q - zs33 * m_qdelta;
m_e1 = m_k + 4.0 * m_g / 3.0
m_e2 = m_k - 2.0 * m_g / 3.0
m_sh2 = 2.0 * m_g
m_x1 = m_e1 - m_e2 * m_npsi + (m_e1 * m_npsi - m_e2) * m_nphi
if abs(m_x1) < 1e-6 * (abs(m_e1) + abs(m_e2)) then
  $sm_err = 5
  nerr = 126
  error = 1
end_if
; --- set tension to prism apex if larger than apex ---
$apex = m_ten
if m_fric # 0.0 then
  $apex = m_csnp / (m_nphi - 1)
end_if
m_ten = min($apex, m_ten)
;
Case 2
; -----
; Running section
; -----
zvise = 1.0
if m_ind # 0.0 then
  m_ind = 2.0
end_if
$anphi = m_nphi
; --- get new trial stresses from old, assuming elastic increments ---
$s11i = zs11 + (zde22 + zde33) * m_e2 + zde11 * m_e1
$s22i = zs22 + (zde11 + zde33) * m_e2 + zde22 * m_e1
$s12i = zs12 + zde12 * m_sh2
$s33i = zs33 + (zde11 + zde22) * m_e2 + zde33 * m_e1
$sdif = $s11i - $s22i
$s0 = 0.5 * ($s11i + $s22i)
$rad = 0.5 * sqrt($sdif * $sdif + 4.0 * $s12i * $s12i)
; --- principal stresses ---
$s1 = $s0 - $rad
$s2 = $s0 + $rad
$psdif = $s1 - $s2
; --- determine case ---
section
if $s33i > $s2i then
  ; --- s33 is major p.s. ---
  $scase = 3
  $s1 = $s1
  $s2 = $s2
  $s3 = $s33i
  exit section
end_if
if $s33i < $s1i then
  ; --- s33 is minor p.s. ---
  $scase = 2
  $s1 = $s33i
  $s2 = $s1
  $s3 = $s2i
  exit section
end_if
; --- s33 is intermediate ---
$scase = 1
$s1 = $s1
$s2 = $s33i
$s3 = $s2i
end_section
;
section
; --- shear yield criterion ---
$fs = $s1 - $s3 * $anphi + m_csnp
$salam = 0.0
; --- tensile yield criterion ---
$ft = m_ten - $s3
$salamt = 0.0
; --- tests for failure ---
if $ft < 0.0 then
  $bisc = sqrt(1.0 + $anphi * $anphi) + $anphi
  $pdiv = -$ft + ($s1 - $anphi * m_ten + m_csnp) * $bisc
  if $pdiv < 0.0 then
    ; --- shear failure ---
    $salam = $fs / m_x1
    $s1 = $s1 - $salam * (m_e1 - m_e2 * m_npsi)
    $s2 = $s2 - $salam * m_e2 * (1.0 - m_npsi)
    $s3 = $s3 - $salam * (m_e2 - m_e1 * m_npsi)
    m_ind = 1.0
  else
    ; --- tension failure ---
    $salamt = $ft / m_e1
    $stco = $salamt * m_e2
    $s1 = $s1 + $stco
    $s2 = $s2 + $stco
    $s3 = m_ten
    m_ind = 3.0
  end_if
else
  if $fs < 0.0 then
    ; --- shear failure ---
    $salam = $fs / m_x1
    $s1 = $s1 - $salam * (m_e1 - m_e2 * m_npsi)
    $s2 = $s2 - $salam * m_e2 * (1.0 - m_npsi)

```

```

    $s3 = $s3 - $salams * (m_e2 - m_e1 * m_npsi)
    m_ind = 1.0
  else
; --- no failure ---
    zs11 = $s11i
    zs22 = $s22i
    zs33 = $s33i
    zs12 = $s12i
    exit section
  end_if
end_if
; --- direction cosines ---
if $psdif = 0.0 then
  $cs2 = 1.0
  $si2 = 0.0
else
  $cs2 = $sdif / $psdif
  $si2 = 2.0 * $s12i / $psdif
end_if
; --- resolve back to global axes ---
case_of $icase
case 1
  $dc2 = ($s1 - $s3) * $cs2
  $dss = $s1 + $s3
  zs11 = 0.5 * ($dss + $dc2)
  zs22 = 0.5 * ($dss - $dc2)
  zs12 = 0.5 * ($s1 - $s3) * $si2
  zs33 = $s2
case 2
  $dc2 = ($s2 - $s3) * $cs2
  $dss = $s2 + $s3
  zs11 = 0.5 * ($dss + $dc2)
  zs22 = 0.5 * ($dss - $dc2)
  zs12 = 0.5 * ($s2 - $s3) * $si2
  zs33 = $s1
case 3
  $dc2 = ($s1 - $s2) * $cs2
  $dss = $s1 + $s2
  zs11 = 0.5 * ($dss + $dc2)
  zs22 = 0.5 * ($dss - $dc2)
  zs12 = 0.5 * ($s1 - $s2) * $si2
  zs33 = $s3
end_case
zvisc = 0.0
; --- accumulate hardening parameter increments ---
if m_ind = 1.0 then
  $de1ps = $salams
  $de3ps = -$salams * m_npsi
  $depm = ($de1ps + $de3ps) / 3.0
  $de1ps = $de1ps - $depm
  $de3ps = $de3ps - $depm
  $seps =
  $seps + sqrt(0.5 * ($de1ps * $de1ps + $depm * $depm + $de3ps * $de3ps))
end_if
if m_ind = 3.0 then
  $sept = $sept - $salamt
end_if
end_section

$sepsav = 0.0
$stpsav = 0.0
if zsub > 0.0 then
  $sepsav = $seps / zsub
  $stpsav = $sept / zsub
; --- reset for the next zone ---
  $seps = 0.0
  $sept = 0.0
end_if
; --- softening/hardening ---
if $sepsav > 0.0 then
  $sepss = m_epdev + $sepsav
  $sftab = int(m_ftab)
  $sidtab = int(m_dtab)
  $sibtab = int(m_btab)
  $sqtat = int(m_qtab)
  if $sftab # 0 then
    m_fric = table($sftab, $sepss)
  end_if
  if $sibtab # 0 then
    m_beta = table($sibtab, $sepss)
  end_if
  if $sidtab # 0 then
    m_delta = table($sidtab, $sepss)
  end_if
  if $sqtat # 0 then
    m_q = table($sqtat, $sepss)
  end_if
; --- data check ---
  $m_err = 0
  if m_fric > 89.0 then
    $m_err = 1
  end_if
  if m_beta < - 1.0 then
    $m_err = 3
  if m_d < 0.0 then
    $m_err = 4
  end_if
  end_if
  if $m_err # 0 then
    nerr = 126
    error = 1
  end_if
;
  m_epdev = $sepss
;
  $sphi = sin(m_fric * degrad)
  $sdelta = sin(m_delta * degrad)
  $sbeta = sin(m_beta * degrad)
  $spsi = sin(m_dil * degrad)
  m_npsi = (1.0 + $spsi) / (1.0 - $spsi)
  m_nphi = (1.0 + $sbeta) / (1.0 - $sphi)
  m_qdelta = $sdelta / (1.0 - $sphi)
  m_csnp = m_q - zs33 * m_qdelta
  m_x1 = m_e1 - m_e2 * m_npsi + (m_e1 * m_npsi - m_e2) * m_nphi
;
  if abs(m_x1) < 1e-6 * (abs(m_e1) + abs(m_e2)) then
    $m_err = 5
    nerr = 126
    error = 1
  end_if
; --- reset tension to prism apex if larger than apex ---
  $apex = m_ten
  if m_fric # 0.0 then
    $apex = m_csnp / (m_nphi - 1)
  end_if
  m_ten = min($apex, m_ten)
end_if
if $stpsav > 0.0 then
  $sepss = m_epten + $stpsav
  $ittab = int(m_ttab)
  if $ittab # 0 then
    m_ten = table($ittab, $sepss)
  end_if
  m_epten = $sepss
  if m_ten < 0.0 then
    $m_err = 4
    nerr = 126
    error = 1
  end_if
end_if

Case 3
; -----
; Return maximum modulus
; -----
  cm_max = m_k + 4.0 * m_g / 3.0
  sm_max = m_g
Case 4
; -----
; Add thermal stresses
; -----
  ztsa = ztea * m_k
  ztsb = zteb * m_k
  ztsc = ztec * m_k
  ztsd = zted * m_k
End_case
end
set echo=on

```